
LetMe is an authorization library for Elixir.
It aims to give you an easy, readable and flexible way of defining authorization
rules by defining a simple DSL, while also giving you introspection functions
that allow you to answer questions like:
	Which actions are defined in my application?
	What are the conditions for a certain action?
	Which actions are allowed for a user with a certain role?

Installation
Add LetMe to mix.exs :
def deps do
  [
    {:let_me, "~> 1.1.0"}
  ]
end
Add LetMe to .formatter.exs:
[
  import_deps: [:let_me]
]
Policy module
The heart of LetMe is the Policy module. It provides a set of macros for
defining the authorization rules of your application. These rules are compiled
to authorization and introspection functions.
A policy module for a simple article CRUD interface might look like this:
defmodule MyApp.Policy do
  use LetMe.Policy

  object :article do
    # Creating articles is allowed if the user role is `editor` or `writer`.
    action :create do
      allow role: :editor
      allow role: :writer
    end

    # Viewing articles is always allowed, unless the user is banned.
    action :read do
      allow true
      deny :banned
    end

    # Updating an article is allowed if (the user role is `editor`) OR
    # (the user role is `writer` AND the article belongs to the user).
    action :update do
      allow role: :editor
      allow [:own_resource, role: :writer]
    end

    # Deleting an article is allowed if the user is an editor.
    action :delete do
      allow role: :editor
    end
  end
end
Whether you want to have a single policy module for your whole application,
or one per context, or whether you want to break it up in any other way is up
to you.
Check module
In general, authorization rules are based on the subject (usually the current
user), the object on which the action is performed, and the action (verb).
LetMe does not make any assumptions on the authorization model or check
implementation.
The checks passed to LetMe.Policy.allow/1 reference functions in the check
module (by default __MODULE__.Checks, so in the example
MyApp.Policy.Checks). Each check function must take the subject, the object,
and optionally an additional argument, and must return a boolean value.
The check module for the policy module above might look like this:
defmodule MyApp.Policy.Checks do
  alias MyApp.Accounts.User

  @doc """
  Returns `true` if the `banned` flag is set on the user.
  """
  def banned(%User{banned: banned}, _, _), do: banned

  @doc """
  Checks whether the user ID of the object matches the ID of the current user.

  Assumes that the object has a `:user_id` field.
  """
  def own_resource(%User{id: id}, %{user_id: id}, _opts) when is_binary(id), do: true
  def own_resource(_, _, _), do: false

  @doc """
  Checks whether the user role matches the role passed as an option.

  ## Usage

      allow role: :editor

  or

      allow {:role, :editor}
  """
  def role(%User{role: role}, _object, role), do: true
  def role(_, _, _), do: false
end
Callbacks
With use LetMe.Policy at the top of your policy module, LetMe will generate
several functions for you.
	Authorization functions: See LetMe.Policy.authorize/3,
LetMe.Policy.authorize!/3 and LetMe.Policy.authorize?/3.
	Introspection functions: See LetMe.Policy.list_rules/0,
LetMe.Policy.list_rules/1, LetMe.Policy.get_rule/1 and others.

Authorization
You can use the authorization functions wherever you need to make authorization
decisions, for example in your context module:
defmodule MyApp.Blog do
  alias MyApp.Accounts.User
  alias MyApp.Blog.Article
  alias MyApp.Policy

  def list_articles(%User{} = current_user) do
    with :ok <- Policy.authorize(:article_read, current_user) do
      {:ok, Repo.all(Article)}
    end
  end

  def fetch_article(id, %User{} = current_user) do
    with :ok <- Policy.authorize(:article_read, current_user, id) do
      case Repo.get(Article, id) do
        nil -> {:error, :not_found}
        article -> {:ok, article}
      end
    end
  end

  def create_article(params, %User{} = current_user) do
    with :ok <- Policy.authorize(:article_create, current_user) do
      %Article{}
      |> Article.changeset(params)
      |> Repo.insert()
    end
  end

  def update_article(%Article{} = article, params, %User{} = current_user) do
    with :ok <- Policy.authorize(:article_update, current_user, article) do
      article
      |> Article.changeset(params)
      |> Repo.update()
    end
  end

  def delete_article(%Article{} = article, %User{} = current_user) do
    with :ok <- Policy.authorize(:article_delete, current_user, article) do
      Repo.delete(article)
    end
  end
end
Introspection
With the introspection functions, you can get the complete list of authorization
rules, e.g. to display them in a documentation page:
iex> MyApp.Policy.list_rules()
[
  %LetMe.Rule{
    action: :create,
    allow: [
      [role: :admin],
      [role: :writer]
    ],
    deny: [],
    description: nil,
    name: :article_create,
    object: :article,
    pre_hooks: [],
    metadata: []
  },
  # ...
]
You can also find a rule by its name:
iex> MyApp.Policy.get_rule(:article_create)
%LetMe.Rule{
  action: :create,
  allow: [
    [role: :admin],
    [role: :writer]
  ],
  name: :article_create,
  object: :article,
  # ...
}
Or you can list all actions tied to a certain role (or any other check):
iex> MyApp.Policy.list_rules(allow: {:role, :writer})
[
  %LetMe.Rule{
    action: :create,
    object: :article,
    # ...
  },
  %LetMe.Rule{
    action: :update,
    object: :article,
    # ...
  }
]
You can also define metadata on an action, which can be useful to extend
the functionality of the library.
For example, you may want to expose some of your actions through your Absinthe
GraphQL schema, but exclude certain actions from it. You could solve
this by adding a :gql_exclude key to the metadata.
defmodule GraphqlPolicy do
  use LetMe.Policy

  object :user do
    action :disable do
      allow role: :admin
      metadata :gql_exclude, true
    end
  end
end
iex> MyApp.Policy.get_rule(:user_disable)
%LetMe.Rule{
  action: :disable,
  allow: [
    [role: :admin]
  ],
  deny: [],
  description: nil,
  name: :user_disable,
  object: :user,
  pre_hooks: [],
  metadata: [
    gql_exclude: true
  ]
}
Scoped queries
Even if a user is generally allowed to see a certain resource type, they may
only be allowed to see a subset of the data. For example, in a blog system, a
user might only be allowed to see published articles, unless they are a writer.
Or in a system where users belong to companies, a company user might only be
allowed to see users who belong to the same company.
In order to scope your queries depending on the user type, you can implement the
LetMe.Schema.scope/3 callback of the LetMe.Schema behaviour, usually in
your Ecto schema module.
defmodule MyApp.Blog.Article do
  use Ecto.Schema
  use LetMe.Schema

  import Ecto.Query
  alias MyApp.Accounts.User

  # Ecto schema and changeset

  @impl LetMe.Schema
  def scope(q, user, opts \\ nil)
  def scope(q, %User{role: :editor}, _), do: q
  def scope(q, %User{role: :writer}, _), do: q
  def scope(q, %User{}, _), do: where(q, published: true)
end
Here, the Ecto query is modified to only return published articles, unless the
user is an editor or writer. You can use the third argument for additional
options.
With this, you can then update your list and fetch functions:
def list_articles(%User{} = current_user) do
  with :ok <- Policy.authorize(:article_read, current_user) do
    articles =
      Article
      |> Article.scope(current_user)
      |> Repo.all()

    {:ok, articles}
  end
end

def fetch_article(id, %User{} = current_user) do
  with :ok <- Policy.authorize(:article_read, current_user, id) do
    result =
      Article
      |> where(id: ^id)
      |> Article.scope(current_user)
      |> Repo.one()

    case result do
      nil -> {:error, :not_found}
      article -> {:ok, article}
    end
  end
end
Looks familiar? Yes, this is nearly the same as in
Bodyguard. However, Bodyguard exposes a
Bodyguard.scope/2 function, which does nothing more than to derive the Ecto
schema module from the Ecto.Queryable and dispatch the call to that module.
But since you usually know which schema module you are dealing with, I believe
this is of limited use, and hence you need to call the scope/2 function of
your Ecto schema directly. The behaviour then only serves to enforce the
pattern.
Field redactions
Sometimes a user may be allowed to retrieve a resource, but some fields should
be hidden. For example, a user might be allowed to see some general information
about another user, such as the name and avatar, but should not be able to see
contact information like the email or the phone number. You could handle
situations like this by conditionally showing or hiding certain information in
the frontend, but it would be cleaner if the context functions would not return
those fields in the first place.
To assist in these kinds of situations, the LetMe.Schema behaviour has another
callback: LetMe.Schema.redacted_fields/3.
defmodule MyApp.Accounts.User do
  use Ecto.Schema
  use LetMe.Schema

  alias MyApp.Accounts.User

  # Ecto schema and changeset

  @impl LetMe.Schema
  def redacted_fields(%User{}, %User{role: :admin}, _), do: []
  def redacted_fields(%User{id: id}, %User{id: id}, _), do: []
  def redacted_fields(%User{}, %User{}, _), do: [:email, :phone_number]
end
The redacted_fields/2 function takes the object as the first argument and the
subject as the second argument. The third argument can be used to pass any
additional options. The function returns a list of fields that need to be
redacted.
In the example above, all fields are visible if the user is an admin or if the
user that is viewed (the object) is the same as the current user (the subject).
In all other cases, the email and phone number are hidden.
You have two options to handle field redactions:
	Modify the query to not select the redacted fields.
	Redact the fields after retrieving the resource(s) from the database.

Modifying the query
You can get the non-virtual schema fields with the __schema__/1 function that
Ecto defines in your Ecto module, reject the redacted fields from the list and
then add a select clause with only the unredacted fields.
def list_users(%User{} = current_user) do
  fields = User.__schema__(:fields)
  filtered_fields = LetMe.reject_redacted_fields(fields, %User{}, current_user)

  Article
  |> select(^filtered_fields)
  |> Repo.all()
end
The advantage of this method is that the fields won't even be transferred from
the DB.
The drawback is that you cannot make decisions on which fields to select
based on data in the struct. Used with the redacted_fields/2 function above,
we can make sure that admins can see all fields, but we can not make sure that
users can see all fields of their own user account.
Another drawback is that all redacted fields will be returned as nil, and you
won't be able to distinguish which fields were redacted and which fields are
just empty, which might be information you would want to display in the
frontend.
Also, you might have more complex select clauses that are not compatible with
this syntax.
Redacting the query result
To mitigate these shortcomings, you can do the redactions after retrieving the
data from the database using LetMe.redact/2.
def list_articles(%User{} = current_user) do
  Article
  |> Repo.all()
  |> LetMe.redact(current_user)
end
The redact function can handle structs, lists of structs, and nil values.
Why would I want to use this library?
	You want an easy-to-read DSL for authorization rules that still allows you
to implement your authorization checks in any way you want.
	You prefer to put your authorization rules into your business layer and
decouple them from your interfaces.
	You prefer to keep your authorization rules in one place (or one place per
context, or similar).
	You want to generate a list of authorization rules.
	You want to filter your authorization rules, e.g. to find out which actions
are available to a certain user role.
	You want a library that can also help you with query scopes and field
redactions.
	You want a library with zero dependencies.

Why wouldn't I want to use this library?
	You prefer to couple authorization checks to your interfaces.
	You prefer to use plugs or middlewares for authorization checks and want the
necessary plumbing ready for use (you can of course build your own plugs and
middlewares around the functions of this library).
	You don't like DSLs and prefer to write functions (note that the DSL only
describes which checks to run and how to apply them, though; you still
write the actual checks as regular functions).
	You don't care about introspection.
	You want to return details on why an authorization request fails. Checks
in LetMe must currently return a boolean value, which means you'll only be
able to give your users a generic error, without telling them which exact
check failed.

Status
This library is actively maintained, but since it has no dependencies and the
feature set is narrowly scoped, you may not see a whole lot of updates. Should
you miss something, though, don't hesitate to open an issue.
Alternatives
For comparison, please have a look at these Elixir libraries:
	Canada
	Canary
	Bodyguard
	Speakeasy

You might also find the article
Authorization for Phoenix Contexts helpful.

Unreleased
[1.1.0] - 2023-05-08
Added
	Added a metadata macro to add metadata to actions. The metadata can be read
from the LetMe.Rule struct.

[1.0.3] - 2023-03-21
Changed
	Update ex_doc and other dev dependencies.

[1.0.2] - 2023-01-05
Added
	Added a cheat sheet for rules and checks.

Fixed
	Fixed a code example for rule introspection in the readme.

[1.0.1] - 2022-11-06
Changed
	Use Keyword.pop/3 with default value instead of Keyword.pop!/2, so that
you can pass options to LetMe.redact/3 without passing the redact_value
option.

[1.0.0] - 2022-11-06
Added
	Added LetMe.Policy.filter_allowed_actions/3 and
LetMe.filter_allowed_actions/4.
	Added LetMe.Policy.get_object_name/1.

Changed
	Renamed c:LetMe.Policy.authorized?/3 to LetMe.Policy.authorize?/3,
because consistency is more important than grammar, maybe.
	The c:LetMe.Schema.scope/2 callback was removed in favour of
LetMe.Schema.scope/3. The __using__ macro defined default
implementations for both functions that returned the given query unchanged, in
case you only needed the redact callback of the behaviour. In practice, this
made it all too easy to call the 2-arity version when only the 3-arity
version was defined, and vice versa, which would lead the query to not be
scoped. So in order to reduce the room for error at the cost of a minor
inconvenience, you will now always need to implement the 3-arity function,
even if you don't need the third argument.
	Changed c:LetMe.Schema.redacted_fields/2 to
LetMe.Schema.redacted_fields/3 to allow passing additional options, and to
be consistent with LetMe.Schema.scope/3.

[0.2.0] - 2022-07-12
Changed
	Added support for nested field redactions, either by explicitly listing the
fields or by referencing a module that also implements LetMe.Schema.

Fixed
	reject_redacted_fields/3 called redact/2 callback with the wrong argument
order.

[0.1.0] - 2022-07-11
initial release

Minimal example
Policy definition
Policy module
defmodule MyApp.Policy do
  use LetMe.Policy

  object :article do
    action :create do
      allow role: :writer
    end
  end
end
Check module
defmodule MyApp.Policy.Checks do
  def role(%MyApp.User{role: role}, _object, role), do: true
  def role(_, _, _), do: false
end
Context module
defmodule MyApp.Blog do
  alias MyApp.Blog.Article
  alias MyApp.Policy

  def create_article(params, %MyApp.User{} = current_user) do
    with :ok <- Policy.authorize(:article_create, current_user) do
      %Article{}
      |> Article.changeset(params)
      |> Repo.insert()
    end
  end
end
Check examples
Check without options
Policy module
object :article do
  action :create do
    allow :writer
  end
end
Check implementation
def writer(%MyApp.User{role: :writer}, _object, _opts), do: true
def writer(_, _, _), do: false
Check with options
Policy module
object :article do
  action :delete do
    allow trust_level: 50
  end
end
Check implementation
def trust_level(%MyApp.User{trust_level: actual_level}, _, required_level)
  when actual_level >= required_level,
  do: true

def trust_level(_, _, _), do: false
Check that depends on the object
Policy module
object :article do
  action :update do
    allow :own_resource
  end
end
Check implementation
def own_resource(%MyApp.User{id: user_id}, %{user_id: user_id}, _), do: true
def own_resource(_, _, _), do: false
Rule examples
Multiple actions with the same rules
object :article do
  action [:create, :update, :delete] do
    allow :admin
  end
end
Combine checks with AND
action :create do
  allow [:two_fa_enabled, role: :writer]
end
Combine checks with OR
action :create do
  allow role: :admin
  allow role: :writer
end
Conditionally allow with exception
action :create do
  allow :is_admin
  deny :is_suspended
end
Always allow
action :read do
  allow true
end
Always deny
action :read do
  deny true
end
Always allow with exception
action :read do
  allow true
  deny :user_is_suspended
end
Add description
action :create do
  desc "allows a user to create a new article"
  allow role: :writer
end
Pre-hooks
Without options
Policy module
object :article do
  action :create do
    pre_hooks :preload_roles

    allow role: :admin
    allow role: :editor
    allow role: :writer
  end
end
Check module
def role(%MyApp.User{roles: roles}, _object, role) do
  Enum.any?(roles, & &1.id == role)
end

def role(_, _, _), do: false

def preload_roles(subject, object) do
  {MyApp.Repo.preload(subject, [:roles]), object}
end
With options
Policy module
object :article do
  action :create do
    pre_hooks {:preload_roles, force: true}

    allow role: :admin
    allow role: :editor
    allow role: :writer
  end
end
Check module
def role(%MyApp.User{roles: roles}, _object, role) do
  Enum.any?(roles, & &1.id == role)
end

def role(_, _, _), do: false

def preload_roles(subject, object, opts) do
  {MyApp.Repo.preload(subject, [:roles], opts), object}
end
From a different module
object :article do
  action :create do
    pre_hooks {MyApp.Policy.Prehooks, :preload_roles, force: true}

    allow role: :admin
    allow role: :editor
    allow role: :writer
  end
end
Multiple pre-hooks
object :article do
  action :create do
    pre_hooks [:preload_roles, :role_list_to_role_id_list]

    allow role: :admin
    allow role: :editor
    allow role: :writer
  end
end
Metadata
object :article do
  action :create do
    allow role: :admin
    allow role: :editor
    allow role: :writer
    
    metadata :gql_exclude, true
    metadata :desc_ja, "ユーザーが新しい記事を作成できるようにする"
  end
end

LetMe is library for defining and evaluating authorization rules and handling
query scopes and field redactions.
This module only defines auxiliary functions. The main functionality lies in
the LetMe.Policy module.

    Functions
  


    
      
        filter_allowed_actions(rules, subject, object, policy)

      


        Takes a list of rules and only returns the rules that would evaluate to true
for the given subject and object.



    


    
      
        filter_rules(rules, opts)

      


        Takes a list of rules and a list of filter options and returns a filtered
list of rules.



    


    
      
        redact(struct, subject, opts \\ [])

      


        Takes a struct or a list of structs and redacts fields depending on the
subject (user).



    


    
      
        reject_redacted_fields(fields, object, subject, opts \\ [])

      


        Removes redacted fields from a given list of fields.

      Link to this function
    
    filter_allowed_actions(rules, subject, object, policy)


      
       
       View Source
     


  


  

      

          @spec filter_allowed_actions([LetMe.Rule.t()], subject, object, module()) :: [
  LetMe.Rule.t()
]
when subject: any(), object: {atom(), any()} | struct()


      


Takes a list of rules and only returns the rules that would evaluate to true
for the given subject and object.
The object needs to be passed as a tuple, where the first element is the
object name, and the second element is the actual object, e.g.
{:article, %Article{}}.
This function is used internally by LetMe.Policy.filter_allowed_actions/3.

  
  example

  
  Example


rules = MyApp.Policy.list_rules()

filter_allowed_actions(
  rules,
  %User{},
  {:article, %Article{}},
  MyApp.Policy
)

  



  
    
      
      Link to this function
    
    filter_rules(rules, opts)


      
       
       View Source
     


  


  

Takes a list of rules and a list of filter options and returns a filtered
list of rules.
This function is used by LetMe.Policy.list_rules/1.

  
  filter-options

  
  Filter options


	:object - Matches an object exactly.
	:action - Matches an action exactly.
	:allow - Either a check name as an atom or a 2-tuple with the check name
and the options.
	:deny - Either a check name as an atom or a 2-tuple with the check name
and the options.

If an atom is passed as allow or deny, the atom is interpreted as a check
name and all rules using the given check name are returned, regardless of
whether additional options are passed to the check. If a 2-tuple is passed,
the first tuple element must be the check name as an atom and the second
tuple element must be the check options. In this case, all rules are returned
that use the given check with exactly the same options. In either case, rules
that have more checks in addition to the given one will also be returned.

  
  examples

  
  Examples


iex> rules = [
...>   %LetMe.Rule{action: :create, name: :article_create, object: :article},
...>   %LetMe.Rule{action: :create, name: :category_create, object: :category}
...> ]
iex> filter_rules(rules, object: :article)
[%LetMe.Rule{action: :create, name: :article_create, object: :article}]

iex> rules = [
...>   %LetMe.Rule{
...>     action: :create,
...>     name: :article_create,
...>     object: :article,
...>     allow: [[role: :editor]]
...>   },
...>   %LetMe.Rule{
...>     action: :update,
...>     name: :article_update,
...>     object: :article,
...>     allow: [:own_resource, [role: :writer]]
...>   }
...> ]
iex> filter_rules(rules, allow: :own_resource)
[%LetMe.Rule{action: :update, name: :article_update, object: :article, allow: [:own_resource, [role: :writer]]}]
iex> match?([_, _], filter_rules(rules, allow: :role))
true
iex> filter_rules(rules, allow: {:role, :editor})
[%LetMe.Rule{action: :create, name: :article_create, object: :article, allow: [[role: :editor]]}]
iex> filter_rules(rules, allow: {:role, :writer})
[%LetMe.Rule{action: :update, name: :article_update, object: :article, allow: [:own_resource, [role: :writer]]}]

  



    

  
    
      
      Link to this function
    
    redact(struct, subject, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec redact(struct(), any(), keyword()) :: struct()


          @spec redact([struct()], any(), keyword()) :: [struct()]


          @spec redact(nil, any(), keyword()) :: nil


      


Takes a struct or a list of structs and redacts fields depending on the
subject (user).
Uses the callback implementation for LetMe.Schema.redacted_fields/3 in the
struct module.

  
  options

  
  Options


	:redact_value - The value to be used for redacted fields. Defaults to
:redacted.

Any additional options will be passed to LetMe.Schema.redacted_fields/3.

  
  example

  
  Example


iex> article = %MyApp.Blog.Article{}
iex> user = %{id: 2, role: :user}
iex> redact(article, user)
%MyApp.Blog.Article{like_count: :redacted, title: "Give us back our moon dust and cockroaches", user_id: 1, view_count: :redacted}

iex> article = %MyApp.Blog.Article{}
iex> user = %{id: 2, role: :user}
iex> redact(article, user, redact_value: nil)
%MyApp.Blog.Article{like_count: nil, title: "Give us back our moon dust and cockroaches", user_id: 1, view_count: nil}

iex> articles = [
...>   %MyApp.Blog.Article{},
...>   %MyApp.Blog.Article{user_id: 2, title: "Joey Chestnut is chomp champ"}
...> ]
iex> user = %{id: 2, role: :user}
iex> redact(articles, user)
[%MyApp.Blog.Article{like_count: :redacted, title: "Give us back our moon dust and cockroaches", user_id: 1, view_count: :redacted}, %MyApp.Blog.Article{like_count: 25, title: "Joey Chestnut is chomp champ", user_id: 2, view_count: :redacted}]

  



    

  
    
      
      Link to this function
    
    reject_redacted_fields(fields, object, subject, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec reject_redacted_fields([atom()], struct(), any(), keyword()) :: [atom()]


      


Removes redacted fields from a given list of fields.
Uses the LetMe.Schema.redacted_fields/3 callback implementation of the
struct module to determine the fields to remove.

  
  examples

  
  Examples


iex> fields = [:like_count, :title, :user_id, :view_count]
iex> user = %{id: 1, role: :user}
iex> article = %MyApp.Blog.Article{}
iex> reject_redacted_fields(fields, article, user)
[:like_count, :title, :user_id]
This can be useful as a safeguard to prevent accidentally casting fields the
user is not allowed to see and thereby nilifying or replacing them.
def update_changeset(%Article{} = article, attrs, %User{} = user) do
  fields = LetMe.reject_redacted_fields(
    [:title, :body, :internal_reference],
    article,
    user
  )

  article
  |> cast(attrs, fields)
  |> validate_required([:title, :body])
end
If a keyword list is given as a fourth argument, it is passed to
LetMe.Schema.redacted_fields/3.

This module defines a DSL for authorization rules and compiles these rules
to authorization and introspection functions.
Usage
defmodule MyApp.Policy do
  use LetMe.Policy

  object :article do
    # Creating articles is allowed if the user role is `editor` or `writer`.
    action :create do
      allow role: :editor
      allow role: :writer
    end

    # Viewing articles is always allowed, unless the user is banned.
    action :read do
      allow true
      deny :banned
    end

    # Updating an article is allowed if (the user role is `editor`) OR
    # (the user role is `writer` AND the article belongs to the user).
    action :update do
      allow role: :editor
      allow [:own_resource, role: :writer]
    end

    # Deleting an article is allowed if the user is an editor.
    action :delete do
      allow role: :editor
    end
  end
end
Options
These options can be passed when using this module:
	check_module - The module where the check functions are defined. Defaults
to __MODULE__.Checks.
	error_reason - The error reason used by the authorize/3 callback.
Defaults to :unauthorized.

Check module
The checks passed to allow/1 and deny/1 reference the names of functions
in the check module.
By default, LetMe tries to find the functions in __MODULE__.Checks (in the
example, this would be MyApp.Policy.Checks). However, you can override the
default check module:
use LetMe.Policy, check_module: MyApp.AuthChecks
Each check function has to take the subject (user), the object, and optionally
an additional argument, and must return a boolean value.
For example, this check determines whether a user is banned:
def banned(%User{banned: true}, _), do: true
def banned(%User{}, _), do: false
This check determines whether the user has the given role:
def role(%User{role: role}, _, role), do: true
def role(_, _, _), do: false
And this check determines whether the object belongs to the user:
def own_resource(%User{id: user_id}, %{user_id: user_id}), do: true
def own_resource(_, _), do: false
LetMe does not make any assumptions about your access control model, as long
as you can map your rules to subject, object and action. You can use the three
rules above with the allow/1 and deny/1 macros.
allow role: :admin
allow :own_resource
deny :banned
Combining checks
Rules evaluate to false by default. These rules will always be false
because they don't have any allow clauses:
action :create do
end

action :update do
  deny false
end
Trying to evaluate a rule name that does not exist also evaluates to false.
As soon as one deny check evaluates to true, the whole rule will evaluate
to false. This rule will always evaluate to false:
action :create do
  allow true
  deny true
end
If you pass a list of checks to either allow/1 or deny/1, the checks
are combined with a logical AND.
# false
action :create do
  allow [true, false]
end

# true
action :create do
  allow [true, true]
end

# true
action :create do
  allow [true, true]
  deny [true, false]
end

# false
action :create do
  allow [true, true]
  deny [true, true]
end
On the other hand, if either the allow/1 or the deny/1 macro is used
multiple times, the checks are combined with a logical OR.
# true
action :create do
  allow true
  allow false
end

# false
action :create do
  allow [true, false]
  allow false
end

# true
action :create do
  allow [true, false]
  allow true
end

# false
action :create do
  allow [true, true]
  allow true
  deny false
  deny true
end
Pre-hooks
You can use pre-hooks to process or gather additional data about the subject
and/or object before running the checks. This can be useful if you need to
preload associations or make external requests. Pre-hooks run once per
authorization request before running the checks. See the documentation for
pre_hooks/1.

    Callbacks
  


    
      
        authorize(atom, any, any)

      


        Authorizes a request defined by the action, subject and object.



    


    
      
        authorize!(atom, any, any)

      


        Same as authorize/3, but raises an error if unauthorized.



    


    
      
        authorize?(atom, any, any)

      


        Same as authorize/3, but returns a boolean.



    


    
      
        fetch_rule(atom)

      


        Returns the rule for the given name. Returns an :ok tuple or :error.



    


    
      
        fetch_rule!(atom)

      


        Returns the rule with the given name. Raises if the rule is not found.



    


    
      
        filter_allowed_actions(list, subject, object)

      


        Takes a list of rules and only returns the rules that would evaluate to true
for the given subject and object.



    


    
      
        get_object_name(module)

      


        Returns the object name for the given schema module or struct, if it was
registered using object/3.



    


    
      
        get_rule(atom)

      


        Returns the rule for the given rule name. Returns nil if the rule is
not found.



    


    
      
        get_schema(atom)

      


        Returns the schema module for the given object name, if it was registered
using object/3.



    


    
      
        list_rules()

      


        Returns all authorization rules as a list.



    


    
      
        list_rules(keyword)

      


        Same as list_rules/0, but takes a keyword list with filter options.



    





  
    Functions
  


    
      
        action(names, list)

      


        Defines an action that needs to be authorized.



    


    
      
        allow(checks)

      


        Defines the checks to be run to determine if an action is allowed.



    


    
      
        deny(checks)

      


        Defines the checks to be run to determine if an action is denied.



    


    
      
        desc(text)

      


        Allows you to add a description to a rule.



    


    
      
        metadata(key, value)

      


        Assigns metadata to the action in the form of a key value pair.



    


    
      
        object(name, module \\ nil, list)

      


        Defines an object on which actions can be performed.



    


    
      
        pre_hooks(hooks)

      


        Registers one or multiple functions to run in order to hydrate the subject
and/or object of the request.

      Link to this callback
    
    authorize(atom, any, any)


      
       
       View Source
     


  


  

      

          @callback authorize(atom(), any(), any()) :: :ok | {:error, any()}


      


Authorizes a request defined by the action, subject and object.

  
  example

  
  Example


Assume we defined this authorization rule:
object :article do
  action :update do
    allow :own_resource
  end
end
And the :own_resource check is defined as:
def own_resource(%{id: user_id}, %{user_id: user_id}), do: true
def own_resource(_, _), do: false
The rule name consists of the object and the action name, in this case
:article_create. To authorize the action, we need to pass the rule name, the
subject (current user) and the object (the article to be updated).
iex> article = %{id: 80, user_id: 1}
iex> user_1 = %{id: 1}
iex> user_2 = %{id: 2}
iex> MyApp.Policy.authorize(:article_update, user_1, article)
:ok
iex> MyApp.Policy.authorize(:article_update, user_2, article)
{:error, :unauthorized}
If the checks don't require the object, it can be omitted.
object :user do
  action :list do
    allow {:role, :admin}
    allow {:role, :client}
  end
end

iex> user = %{id: 1, role: :admin}
iex> MyApp.Policy.authorize(:user_list, user)
:ok
iex> user = %{id: 2, role: :user}
iex> MyApp.Policy.authorize(:user_list, user)
{:error, :unauthorized}
The error reason can be customized by setting the :error_reason option when
using the module.

  



  
    
      
      Link to this callback
    
    authorize!(atom, any, any)


      
       
       View Source
     


  


  

      

          @callback authorize!(atom(), any(), any()) :: :ok


      


Same as authorize/3, but raises an error if unauthorized.

  
  example

  
  Example


With the same authorization rules as defined in the authorize/3
documentation, we get this:
iex> article = %{id: 80, user_id: 1}
iex> user_1 = %{id: 1}
iex> user_2 = %{id: 2}
iex> MyApp.Policy.authorize!(:article_update, user_1, article)
:ok
iex> MyApp.Policy.authorize!(:article_update, user_2, article)
** (LetMe.UnauthorizedError) unauthorized

  



  
    
      
      Link to this callback
    
    authorize?(atom, any, any)


      
       
       View Source
     


  


  

      

          @callback authorize?(atom(), any(), any()) :: boolean()


      


Same as authorize/3, but returns a boolean.

  
  example

  
  Example


With the same authorization rules as defined in the authorize/3
documentation, we get this:
iex> article = %{id: 80, user_id: 1}
iex> user_1 = %{id: 1}
iex> user_2 = %{id: 2}
iex> MyApp.Policy.authorize?(:article_update, user_1, article)
true
iex> MyApp.Policy.authorize?(:article_update, user_2, article)
false

  



  
    
      
      Link to this callback
    
    fetch_rule(atom)


      
       
       View Source
     


  


  

      

          @callback fetch_rule(atom()) :: {:ok, LetMe.Rule.t()} | :error


      


Returns the rule for the given name. Returns an :ok tuple or :error.
The rule name is an atom with the format {object}_{action}.

  
  example

  
  Example


iex> MyApp.Policy.fetch_rule(:article_create)
{:ok,
 %LetMe.Rule{
   action: :create,
   allow: [[role: :admin], [role: :writer]],
   deny: [],
   name: :article_create,
   object: :article,
   pre_hooks: []
 }}

 iex> MyApp.Policy.fetch_rule(:cookie_eat)
 :error

  



  
    
      
      Link to this callback
    
    fetch_rule!(atom)


      
       
       View Source
     


  


  

      

          @callback fetch_rule!(atom()) :: LetMe.Rule.t()


      


Returns the rule with the given name. Raises if the rule is not found.
The rule name is an atom with the format {object}_{action}.

  
  example

  
  Example


iex> MyApp.Policy.fetch_rule!(:article_create)
%LetMe.Rule{
  action: :create,
  allow: [[role: :admin], [role: :writer]],
  deny: [],
  name: :article_create,
  object: :article,
  pre_hooks: []
}

  



  
    
      
      Link to this callback
    
    filter_allowed_actions(list, subject, object)


      
       
       View Source
     


  


  

      

          @callback filter_allowed_actions([LetMe.Rule.t()], subject, object) :: [LetMe.Rule.t()]
when subject: any(), object: {atom(), any()} | struct()


      


Takes a list of rules and only returns the rules that would evaluate to true
for the given subject and object.

  
  examples

  
  Examples


The object can be passed as a tuple, where the first element is the
object name, and the second element is the actual object, e.g.
{:article, %Article{}}.
iex> rules = MyApp.Policy.list_rules()
iex> MyApp.Policy.filter_allowed_actions(
...>   rules,
...>   %{id: 2, role: nil},
...>   {:article, %MyApp.Blog.Article{}}
...> )
[
  %LetMe.Rule{
    action: :view,
    allow: [true],
    deny: [],
    description: "allows to view an article and the list of articles",
    name: :article_view,
    object: :article,
    pre_hooks: []
  }
]
If you registered the schema module with LetMe.Policy.object/3, you can also
pass the schema module or the struct instead of a tuple.
iex> rules = MyApp.Policy.list_rules()
iex> MyApp.Policy.filter_allowed_actions(
...>   rules,
...>   %{id: 2, role: nil},
...>   %MyApp.Blog.Article{}
...> )
[
  %LetMe.Rule{
action: :view,
allow: [true],
deny: [],
description: "allows to view an article and the list of articles",
name: :article_view,
object: :article,
pre_hooks: []
  }
]

  



  
    
      
      Link to this callback
    
    get_object_name(module)


      
       
       View Source
     


  


  

      

          @callback get_object_name(module()) :: atom() | nil


      


Returns the object name for the given schema module or struct, if it was
registered using object/3.

  
  examples

  
  Examples


iex> MyApp.Policy.get_object_name(MyApp.Blog.Article)
:article

iex> MyApp.Policy.get_object_name(%MyApp.Blog.Article{})
:article

iex> MyApp.Policy.get_object_name(MyApp.Blog.Tag)
nil

  



  
    
      
      Link to this callback
    
    get_rule(atom)


      
       
       View Source
     


  


  

      

          @callback get_rule(atom()) :: LetMe.Rule.t() | nil


      


Returns the rule for the given rule name. Returns nil if the rule is
not found.
The rule name is an atom with the format {object}_{action}.

  
  example

  
  Example


iex> MyApp.Policy.get_rule(:article_create)
%LetMe.Rule{
  action: :create,
  allow: [[role: :admin], [role: :writer]],
  deny: [],
  name: :article_create,
  object: :article,
  pre_hooks: []
}

iex> MyApp.Policy.get_rule(:cookie_eat)
nil

  



  
    
      
      Link to this callback
    
    get_schema(atom)


      
       
       View Source
     


  


  

      

          @callback get_schema(atom()) :: module() | nil


      


Returns the schema module for the given object name, if it was registered
using object/3.

  
  examples

  
  Examples


iex> MyApp.Policy.get_schema(:article)
MyApp.Blog.Article

iex> MyApp.Policy.get_schema(:user)
nil

  



  
    
      
      Link to this callback
    
    list_rules()


      
       
       View Source
     


  


  

      

          @callback list_rules() :: [LetMe.Rule.t()]


      


Returns all authorization rules as a list.

  
  example

  
  Example


iex> MyApp.PolicyShort.list_rules()
[
  %LetMe.Rule{
    action: :create,
    allow: [[role: :admin], [role: :writer]],
    deny: [],
    name: :article_create,
    object: :article,
    pre_hooks: []
  },
  %LetMe.Rule{
    action: :update,
    allow: [:own_resource],
    deny: [],
    name: :article_update,
    object: :article,
    pre_hooks: [:preload_groups]
  }
]

  



  
    
      
      Link to this callback
    
    list_rules(keyword)


      
       
       View Source
     


  


  

      

          @callback list_rules(keyword()) :: [LetMe.Rule.t()]


      


Same as list_rules/0, but takes a keyword list with filter options.
See LetMe.filter_rules/2 for a list of available filter options.

      Link to this macro
    
    action(names, list)


      
       
       View Source
     


      (macro)

  


  

      

          @spec action(atom() | [atom()], Macro.t()) :: Macro.t()


      


Defines an action that needs to be authorized.
Within the do-block, you can use the allow/1, deny/1 and pre_hooks/1
macros to define the checks to be run and the desc/1 macro to add a
description.
This macro must be used within the do-block of object/2.
Each action block will be compiled to a rule. The rule name is an atom with
the format {object}_{action}.

  
  example

  
  Example


object :article do
  action :create do
    allow role: :editor
    allow role: :writer
  end

  action :update do
    allow role: :editor
    allow [:own_resource, role: :writer]
  end
end
If you have multiple actions with the same allow and deny rules, you can also
pass a list of action names as the first argument.
object :article do
  action [:create, :update, :delete] do
    allow role: :editor
    allow role: :writer
  end
end

  



  
    
      
      Link to this macro
    
    allow(checks)


      
       
       View Source
     


      (macro)

  


  

      

          @spec allow(LetMe.Rule.check() | [LetMe.Rule.check()]) :: Macro.t()


      


Defines the checks to be run to determine if an action is allowed.
The argument can be:
	a function name as an atom
	a tuple with the function name and an additional argument
	a list of function names or function/argument tuples
	true or false - Always allows or denies an action. Can be useful in
combination with the deny/1 macro.

The function must be defined in the configured check module and take the
subject (current user), object as arguments, and if given, the additional
argument.
If a list is given as an argument, the checks are combined with a logical
AND.
If the allow/1 macro is used multiple times within the same action/2
block, the checks of each macro call are combined with a logical OR.

  
  examples

  
  Examples


Let's assume you defined the following checks:
defmodule MyApp.Policy.Checks do
  def role(%User{role: role}, _, role), do: true
  def role(_, _, _), do: false

  def own_resource(%User{id: id}, %{user_id: id}, _), do: true
  def own_resource(_, _, _), do: false
end
This would allow the :article_update action only if the current user has
the role :admin:
object :article do
  action :update do
    allow role: :admin
  end
end
This is equivalent to:
object :article do
  action :update do
    allow {:role, :admin}
  end
end
This would allow the :article_update action if the user has the role
:writer and the article belongs to the user:
object :article do
  action :update do
    allow [:own_resource, role: :writer]
  end
end
This is equivalent to:
object :article do
  action :update do
    allow [:own_resource, {:role, :writer}]
  end
end
This would allow the :article_update action if
(the user has the role :admin or (the user has the role :writer and
the article belongs to the user)):
object :article do
  action :update do
    allow role: :admin
    allow [:own_resource, role: :writer]
  end
end

  



  
    
      
      Link to this macro
    
    deny(checks)


      
       
       View Source
     


      (macro)

  


  

      

          @spec deny(LetMe.Rule.check() | [LetMe.Rule.check()]) :: Macro.t()


      


Defines the checks to be run to determine if an action is denied.
If any of the checks evaluates to true, the allow checks are overridden
and the authorization request is automatically denied.
If a list is given as an argument, the checks are combined with a logical
AND.
If the allow/1 macro is used multiple times within the same action/2
block, the checks of each macro call are combined with a logical OR.

  
  examples

  
  Examples


Let's assume you defined the following checks:
defmodule MyApp.Policy.Checks do
  def role(%User{role: role}, _, role), do: true
  def role(_, _, _), do: false

  def own_resource(%User{id: id}, %{user_id: id}, _), do: true
  def own_resource(_, _, _), do: false

  def same_user(%User{id: id}, %User{id: id}, _), do: true
  def same_user(_, _, _), do: false
end
This would allow the :user_delete action by default, unless the object is
the current user:
object :user do
  action :delete do
    allow true
    deny :same_user
  end
end
This would allow the :article_update action only if the current user has
the role :admin, unless the object is the current user:
object :user do
  action :delete do
    allow role: :admin
    deny :same_user
  end
end
This would allow the :user_delete by default, unless the object is the
current user and the current user is an admin:
object :user do
  action :delete do
    allow true
    deny [:same_user, role: :admin]
  end
end
This would allow the :user_delete by default, unless the object is the
current user or the current user is a peasant:
object :user do
  action :delete do
    allow true
    deny :same_user
    deny role: :peasant
  end
end

  



  
    
      
      Link to this macro
    
    desc(text)


      
       
       View Source
     


      (macro)

  


  

      

          @spec desc(String.t()) :: Macro.t()


      


Allows you to add a description to a rule.
The description can be accessed from the LetMe.Rule struct. You can use it
to generate help texts or documentation.

  
  example

  
  Example


object :article do
  action :create do
    desc "allows a user to create a new article"
    allow role: :writer
  end
end

  



  
    
      
      Link to this macro
    
    metadata(key, value)


      
       
       View Source
     


      (macro)

  


  

      

          @spec metadata(atom(), term()) :: Macro.t()


      


Assigns metadata to the action in the form of a key value pair.
The metadata can be accessed from the LetMe.Rule struct. You can use it
to extend the functionality of the library.

  
  example

  
  Example


object :article do
  action :create do
    allow role: :writer

    desc "Allows a user to create a new article."
    metadata :desc_ja, "ユーザーが新しい記事を作成できるようにする"
  end
end
The LetMe.Rule struct returned by the introspection functions would look
like this:
%LetMe.Rule{
  action: :create,
  allow: [[role: :writer]],
  deny: [],
  description: "Allows a user to create a new article.",
  name: :article_create,
  object: :article,
  pre_hooks: [],
  metadata: [
    desc_ja: "ユーザーが新しい記事を作成できるようにする"
  ]
}
It is also possible to use the metadata macro multiple times.
object :article do
  action :create do
    allow role: :writer

    desc "Allows a user to create a new article."
    metadata :desc_ja, "ユーザーが新しい記事を作成できるようにする"
    metadata :desc_es, "Permite al usuario crear un nuevo artículo."
  end
end
This would result in:
%LetMe.Rule{
  action: :create,
  allow: [[role: :writer]],
  deny: [],
  description: "Allows a user to create a new article.",
  name: :article_create,
  object: :article,
  pre_hooks: [],
  metadata: [
    desc_ja: "ユーザーが新しい記事を作成できるようにする",
    desc_es: "Permite al usuario crear un nuevo artículo."
  ]
}

  



    

  
    
      
      Link to this macro
    
    object(name, module \\ nil, list)


      
       
       View Source
     


      (macro)

  


  

      

          @spec object(atom(), module() | nil, Macro.t()) :: Macro.t()


      


Defines an object on which actions can be performed.
Within the do-block, you can use the action/2 macro to define the actions
and checks.

  
  examples

  
  Examples


object :article do
  action :create do
    allow role: :writer
  end

  action :delete do
    allow role: :editor
  end
end
You can optionally pass the schema module as the second argument. The schema
module should implement the LetMe.Schema behaviour.
object :article, MyApp.Blog.Article do
  action :create do
    allow role: :writer
  end
end
At the moment, this doesn't do much, except that you can find the schema
module by passing the object name to get_schema/1, or find the object name
by passing the schema module or struct to get_object_name/1 now. Also,
you can now only pass the struct toc:MyApp.Policy.filter_allowed_actions/3,
without explicitly passing the object name.

  



  
    
      
      Link to this macro
    
    pre_hooks(hooks)


      
       
       View Source
     


      (macro)

  


  

      

          @spec pre_hooks(LetMe.Rule.hook() | [LetMe.Rule.hook()]) :: Macro.t()


      


Registers one or multiple functions to run in order to hydrate the subject
and/or object of the request.
This is useful if you need to enhance the data for multiple checks in the
same action by preloading associations, making external requests, or similar
things. The configured hook functions will be called once before running the
checks for an action.
The referenced functions must take the subject and object as arguments and
return a 2-tuple with the updated subject and object.

  
  examples

  
  Examples


Let's assume we defined these check and hook functions in our check module:
def MyApp.Policy.Checks do
  # Checks

  def min_age(%{age: age}, _, min_age), do: age >= min_age

  # Hooks

  def double_age(subject, object) do
    new_subject = %{subject | age: subject.age * 2}
    {new_subject, object}
  end

  def set_age(subject, object, age) do
    new_subject = %{subject | age: age}
    {new_subject, object}
  end
end
If an atom is passed, LetMe will try to find the function in the check module.
object :article do
  action :view do
    pre_hooks :double_age
    allow min_age: 50
  end
end
With this in place, the following authorization request will evaluate to
true:
MyApp.Policy.authorize!(:article_view, %{age: 25})
# => true
If your hooks are defined in a different module, you can also pass a
module/function tuple. The pre-hook configuration above is equivalent to:
object :article do
  action :view do
    pre_hooks {MyApp.Policy.Checks, :double_age}
    allow min_age: 50
  end
end
You can also pass options to a hook by using an MFA tuple:
object :article do
  action :view do
    pre_hooks {MyApp.Policy.Checks, :set_age, 50}
    allow min_age: 50
  end
end

MyApp.Policy.authorize!(:article_view, %{age: 10})
# => true
And finally, you can also pass a list of hooks, which will be run in sequence:
alias MyApp.Policy.Checks

object :article do
  action :view do
    pre_hooks [{Checks, :set_age, 25}, :double_age]
    allow min_age: 50
  end
end

MyApp.Policy.authorize!(:article_view, %{age: 10})
# => true

    Types
  


    
      
        check()

      


        A check references a function in the configured check module.



    


    
      
        hook()

      


        A hook can be registered to hydrate the subject and/or object before passing
them to the check functions.



    


    
      
        t()

      


        Struct for an authorization rule.

      Link to this type
    
    check()


      
       
       View Source
     


  


  

      

          @type check() :: atom() | {atom(), any()}


      


A check references a function in the configured check module.
Can be either one of:
	A function name as an atom. The function must be a 2-arity function that
takes the subject (usually the current user) and the object as
arguments.
	A tuple with the function name as an atom and a value of any type. The
function must be a 3-arity function that takes the subject, the object, and
the given value as arguments.


  



  
    
      
      Link to this type
    
    hook()


      
       
       View Source
     


  


  

      

          @type hook() :: atom() | {module(), atom()} | {module(), atom(), any()}


      


A hook can be registered to hydrate the subject and/or object before passing
them to the check functions.
Can be either one of:
	The name of a function defined in the configured check module as an atom.
	A {module, function} tuple.
	A {module, function, arguments} tuple.

In either case, the function must take the subject as the first argument, the
object as the second argument, and return a tuple with the updated subject and
object. If an MFA tuple is passed, the given arguments are appended to the
default arguments.

  



  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %LetMe.Rule{
  action: atom(),
  allow: [check() | [check()]],
  deny: [check() | [check()]],
  description: String.t() | nil,
  metadata: Keyword.t(),
  name: atom(),
  object: atom(),
  pre_hooks: [hook()]
}


      


Struct for an authorization rule.
	action - The action (verb) to be performed on the object, e.g. :update.
	allow - A list of lists of checks to run to determine whether the action
is allowed. The outer list contains the alternatives (one for each allow
call; combined with OR). The inner lists are the checks for each allow
(combined with AND).
	description - A human-readable description of the action.
	deny - A list of lists of checks to run to determine whether the action is
explicitly denied. Same format as in allow. If any of these checks returns
true, the end result of the authorization request is immediately false,
even if any of the checks in the allow field would return true.
	name - The name of the rule. Is always {object}_{action}.
	object - The object that the action is performed on, e.g. :article.
	pre_hooks - Functions to run in order to hydrate the subject and/or object
before running the allow and deny checks.
	metadata - A list of relevant metadata useful for extending functionality.

The list entries in the outer list of the allow and deny fields are
combined with a logical OR. If one of the entries is a list of checks, those
checks are combined with a logical AND.

  
  examples

  
  Examples


	[{role: :editor}, {role: :writer}] - role is editor OR role is writer
	[[{role: :editor}], [{role: :writer}]] - same as above
	[[{role: :editor}], [{role: :writer}, {:own_resource}]] -
 (role is editor OR (role is writer AND object is the user's own resource))


  


        

      



  

    
LetMe.Schema behaviour
    



      
Defines a behaviour with callbacks for scoping and redactions.
Using this module will define overridable default implementations for the
scope/3 and redacted_fields/3 callbacks.
Usage
defmodule MyApp.Blog.Article do
  use LetMe.Schema
  import Ecto.Schema
  alias MyApp.Accounts.User

  @impl LetMe.Schema
  def scope(q, user, opts \\ nil)
  def scope(q, %User{role: :admin}, _), do: q
  def scope(q, %User{}, _), do: where(q, published: true)

  @impl LetMe.Schema
  def redacted_fields(_, %User{role: :admin}, _), do: []
  def redacted_fields(%__MODULE__{user_id: id}, %User{id: id}, _), do: []
  def redacted_fields(_, %User{}, _), do: [:view_count]
end
Scoping a query
With the setup above, you can scope a blog article query depending on the
user.
defmodule MyApp.Blog do
  import Ecto.Query

  alias MyApp.Accounts.User
  alias MyApp.Blog.Article

  def list_articles(%User{} = current_user) do
    Article
    |> Article.scope(current_user)
    |> Repo.all()
  end

  def get_article(id, %User{} = current_user) when is_integer(id) do
    Article
    |> where(id: id)
    |> Article.scope(current_user)
    |> Repo.one()
  end
end
Redacting fields
After implementing the redacted_fields/3 callback, you can hide fields
from depending on the user by calling LetMe.redact/3 on a struct or a list
of structs.
def list_articles(%User{} = current_user) do
  Article
  |> Repo.all()
  |> LetMe.redact(current_user)
end

    Callbacks
  


    
      
        redacted_fields(object, subject, opts)

      


        Returns the fields that need to be removed from the given object for the given
subject.



    


    
      
        scope(queryable, subject, opts)

      


        Takes a queryable (usually an Ecto.Queryable) and a subject (usually the
current user) and returns an updated queryable.

      Link to this type
    
    redacted_fields()


      
       
       View Source
     


  


  

      

          @type redacted_fields() :: [atom() | {atom(), module() | redacted_fields()}]

      Link to this callback
    
    redacted_fields(object, subject, opts)


      
       
       View Source
     


  


  

      

          @callback redacted_fields(object, subject, opts) :: redacted_fields()
when object: any(), subject: any(), opts: keyword()


      


Returns the fields that need to be removed from the given object for the given
subject.
This function can be used to hide certain fields depending on the current
user. See also LetMe.redact/3 and LetMe.reject_redacted_fields/3.
defmodule MyApp.Blog.Article do
  use LetMe.Schema
  alias MyApp.Accounts.User

  @impl LetMe.Schema
  # hide view count unless the user is an admin or the article was written
  # by the user
  def redacted_fields(_, %User{role: :admin}, _), do: []
  def redacted_fields(%__MODULE__{user_id: id}, %User{id: id}, _), do: []
  def redacted_fields(_, %User{}, _), do: [:view_count]
end
The return value must be a list of fields to redact. You can also redact
nested fields by passing the field names directly:
[:email, sibling: [:phone_number]]
Or you can pass the module name of a nested struct, if that module also
implements LetMe.Schema:
[:email, sibling: MyApp.Relative]
The last argument can be used for any additional options.

  



  
    
      
      Link to this callback
    
    scope(queryable, subject, opts)

          @callback scope(queryable, subject, opts) :: queryable
when queryable: any(), subject: any(), opts: any()


      


Takes a queryable (usually an Ecto.Queryable) and a subject (usually the
current user) and returns an updated queryable.
This allows you to add WHERE clauses to a query depending on the user. For
example, you may want to add a WHERE clause to only return articles that are
published, unless the user is an admin. Or you may want to only return
objects that belong to the user.
defmodule MyApp.Blog.Article do
  use Ecto.Schema
  use LetMe.Schema

  import Ecto.Schema
  alias MyApp.Accounts.User

  # Ecto schema and changeset

  @impl LetMe.Schema
  def scope(q, user, opts \\ nil)
  def scope(q, %User{role: :admin}, _), do: q
  def scope(q, %User{}, _), do: where(q, published: true)
end
Since LetMe does not depend on Ecto and does not make any assumptions on the
queryable passed to the callback function, you are not constrained to use this
mechanism for Ecto queries only. For example, you could use the function to
add filter parameters before passing them to a filter function or making an
API call.
@impl LetMe.Schema
def scope(q, user, opts \\ nil)

def scope(query_params, %User{role: :admin}, _), do: query_params

def scope(query_params, %User{}, _) do
  Keyword.put(query_params, :published, true)
end
You can use the third argument to pass any additional options.

  


        

      



  

    
LetMe.UnauthorizedError exception
    



      
Raised by LetMe.Policy.authorize!/3 if a request is unauthorized.

